code:
init 2 python:
    def student_intern_offer_requirement(the_person):
        if mc.business.college_interns_unlocked and not the_person in mc.business.intern_list:
            if the_person in [lily, erica, kaya, emily]:
                return False
            if not mc.business.has_funds(5000):
                return "$5000 scholarship fund"
            if not len(mc.business.get_intern_depts_with_openings()) > 0:
                return "No internship openings"
            if not the_person.location == university:
                return "Talk on campus"
            return True
        return False

    student_intern_offer_action = Action("Offer internship", student_intern_offer_requirement, "student_intern_offer_label",
        menu_tooltip = "Offer this student an internship with your company.")

init 16 python:
    add_label_hijack("normal_start", "activate_student_role_enhancement")
    add_label_hijack("after_load", "activate_student_role_enhancement")

label student_intern_offer_label(the_person):
    $ scene_manager = Scene()
    $ hated = []
    $ multiplier = 0
    $ scene_manager.add_actor(the_person, position = "sitting")
    "You probably shouldn't just come right out and offer [the_person.title] a $5000 internship, maybe a few questions first."
    mc.name "How is school going [the_person.title]?"
    the_person "It's going really good, I am learning so much."
    if the_person.get_opinion_score("supply work") > 0:
        if the_person.get_opinion_score("supply work") > 1:
            the_person "It is so interesting learning about all the levels of the supply chain and how interconnected everything really is."
        else:
            the_person "We've been practicing cold calls to suppliers and it is so fun playing the person getting the call."
    if the_person.get_opinion_score("HR work") > 0:
        if the_person.get_opinion_score("HR work") > 1:
            the_person "We've been looking at how human resource departments can set the culture of a whole company. It is so cool to think I could have such a big impact."
        else:
            the_person "I'm looking forward to being able to help people with their problems all day."
    if the_person.get_opinion_score("marketing work") > 0:
        if the_person.get_opinion_score("marketing work") > 1:
            the_person "We've been designing ad campaigns, and it is so much fun to really flex my creativity to make people think about a product in a whole new way."
        else:
            the_person "Today we were learning about how much of an impact discounts can have on helping sell a product."
    if the_person.get_opinion_score("production work") > 0:
        if the_person.get_opinion_score("production work") > 1:
            the_person "Learning is great, but I'm so excited to get out there and actually make things with my own hands. I really want to be able to point to something and say I made it."
        else:
            the_person "I'm glad production work is more than sitting at a desk. By the time school is done I'll be ready to stop sitting all day."
    if the_person.get_opinion_score("research work") > 0:
        if the_person.get_opinion_score("research work") > 1:
            the_person "When Professor [nora.last_name] talks about the things she researches I am so excited to get out there and make discoveries of my own."
        else:
            the_person "I'm ready to get out and pursue my own ideas instead of following in the footsteps of my professors."
    mc.name "Wow, that all sounds pretty incredible. I'm glad you are having such a good time in school."
    the_person "Well, to be honest it isn't all great."
    if the_person.get_opinion_score("supply work") < 0:
        the_person "Talking about buying supplies is so boring"
        $ hated.append("supply")
    if the_person.get_opinion_score("HR work") < 0:
        the_person "I'm independent and don't want to have to think about what other people need."
        $ hated.append("HR")
    if the_person.get_opinion_score("marketing work") < 0:
        the_person "I'm never sure what to say when we practice selling things to our classmates."
        $ hated.append("marketing")
    if the_person.get_opinion_score("production work") < 0:
        the_person "I don't really like how physical the production process can be."
        $ hated.append("production")
    if the_person.get_opinion_score("research work") < 0:
        the_person "Recording results of experiments is so boring, and then we repeat them to be safe."
        $ hated.append("research")
    mc.name "Well that is understanding. So you do all that every week?"
    the_person "Pretty much, but if I have to catch up on something I do have my weekends free."
    the_person "Luckily I only have to work during the summer as long as I am careful about how much I spend."
    the_person "Of course that does mean my weekends can be a bit boring, I can't go out partying without a steady income."
    mc.name "That is understandable. You know, if you were willing to sacrifice part of your weekend I might be able to help you afford to have more fun for part of it as well."
    the_person "Really?"
    mc.name "Yeah, I've been working with the university to offer an internship. You'll get some real life work experience and a hefty contribution towards your tuition."
    the_person "That sounds incredible. Is there a catch."
    mc.name "We need you there in the morning and afternoon every Saturday and Sunday, so you'll have to keep your partying from running too late."
    the_person "That's not too bad."
    mc.name "You'll also have to put up with me hanging around, I work most weekends to keep things going."
    the_person "Truly that is a great burden, but I think I'll be able to manage."
    the_person "What would I be doing exactly?"
    "That is a good question. You have five departments at the company but you'll need to find one that she likes and has openings."
    $ score = 0
    if len(hated) > 0:
        $ multiplier = len(hated)
        while multiplier > 0:
            $ the_dept = hated[multiplier-1]
            $ score = the_person.get_opinion_score("[the_dept] work")
            $ the_person.discover_opinion("[the_dept] work")
            if score == -2:
                "[the_person.title] hates [the_dept] work."
            else:
                "[the_person.title] doesn't like [the_dept] work."
            $ multiplier -=1
    menu:
        "Biology (Research)" if len(mc.business.college_interns_research) < mc.business.max_interns_by_division:
            $ the_dept = "Research"
            $ job = student_intern_rd_job
        "Biology \n{color=#ff0000}{size=18}Research Team Full!{/size}{/color} (disabled)" if len(mc.business.college_interns_research) >= mc.business.max_interns_by_division:
            pass
        "Chemistry (Production)" if len(mc.business.college_interns_production) < mc.business.max_interns_by_division:
            $ the_dept = "Production"
            $ job = student_intern_production_job
        "Chemistry \n{color=#ff0000}{size=18}Production Team Full!{/size}{/color} (disabled)" if len(mc.business.college_interns_production) >= mc.business.max_interns_by_division:
            pass
        "Graphic Design (Marketing)" if len(mc.business.college_interns_market) < mc.business.max_interns_by_division and mc.business.college_market_interns_unlocked:
            $ the_dept = "Marketing"
            $ job = student_intern_market_job
        "Graphic Design (Marketing) (disabled)":    #In the future we may have opportunities to recruit interns for these programs.
            pass
        "Psychology (HR)" if len(mc.business.college_interns_HR) < mc.business.max_interns_by_division and mc.business.college_hr_interns_unlocked:
            $ the_dept = "HR"
            $ job = student_intern_hr_job
        "Psychology (HR) (disabled)":
            pass
        "Business (Supply)" if len(mc.business.college_interns_supply) < mc.business.max_interns_by_division and mc.business.college_supply_interns_unlocked:
            $ the_dept = "Supply"
            $ job = student_intern_supply_job
        "Business (Supply) (disabled)":
            pass
        "Never mind":
            $ the_dept = None
    if the_dept:
        mc.name "I've got an opening in the [the_dept] department of my business."
        if the_dept in hated:
            if the_person.get_opinion_score("[the_dept] work") < -1:
                $ multiplier = 2
                the_person "Weren't you listening, I said that I hate [the_dept] work."
            else:
                $ multiplier = 1
                the_person "Weren't you listening, I said that I don't like [the_dept] work."
            menu:
                "Do it for love" if the_person.love > 30+(15*multipler):
                    mc.name "I know, but that is my only opening at the moment. Please, would you do it for me?"
                    the_person "Well... okay, but only because I love you so much."
                    the_person "To be honest I am pretty excited about working with you, even if it means I have to do [the_dept] work."
                "Obey me" if the_person.obedience > 130+(15*multipler):
                    mc.name "Sorry, I misspoke. You are going to work in the [the_dept] department. This isn't a request."
                    the_person "Yes [the_person.mc_title]!"
                    mc.name "That's a good girl."
                    the_person "Of course, anything for you."
                "Do it for money" if mc.business.has_funds(5000*(1+multipler)):
                    mc.name "I know, but sadly that is all I have available at the moment."
                    mc.name "I know it isn't ideal, but would you be willing to do this if I gave you some extra money?"
                    if multiplier == 1:
                        the_person "Okay, but it is going to cost you another $5,000."
                    else:
                        the_person "Really... I guess if you were willing to give me extra $10,000 I would be able to deal with it."
                    menu:
                        "Okay":
                            mc.name "I can make that happen. Let me just wire you the extra money now."
                            "You pull out your phone and get the money transferred, she takes out hers to verify."
                            $ mc.business.change_funds(-5000*multipler)
                            the_person "Looks good."
                        "Never mind":
                            $ the_dept = None
                "Never mind":
                    $ the_dept = None
        else:
            $ the_person.discover_opinion("[the_dept] work")
            if the_person.get_opinion_score("[the_dept] work") < 1:
                the_person "Okay, that's fine."
            elif the_person.get_opinion_score("[the_dept] work") < 2:
                the_person "Great! That sounds like it could be fun."
            else:
                the_person "Really? That is awesome! It is exactly what I was hoping for."
    if the_dept:
        mc.name "Great, I'll get the paperwork all taken care of and then see you this weekend."
        $ mc.business.hire_college_intern(the_person, job)
        $ mc.business.change_funds(-5000)
    else:
        mc.name "Sorry, this isn't going to work. I don't have anything that you will be happy with right now."
        the_person "Oh, well alright. I guess let me know if something changes."
    $ del score
    $ del multiplier
    $ del hated
    $ del the_dept
    $ scene_manager.clear_scene()
    return

label activate_student_role_enhancement(stack):
    python:
        generic_student_role.add_action(student_intern_offer_action)
        execute_hijack_call(stack)
    return


code:
init 0 python:
    def hypno_lactation_on_turn(person):
        person.event_triggers_dict["hypno_lactated_recently"] = False
        if person.event_triggers_dict.get("hypno_lactation_count", 0) > 0:
            if renpy.random.randint(0,10) < person.event_triggers_dict.get("hypno_lactation_count", 0):
                person.event_triggers_dict["hypno_lactation_count"] -= 1
                if person.lactation_sources > 0:
                    person.lactation_sources -= 1
        return

    def trigger_hypno_lactation_requirement(person):
        if person.event_triggers_dict.get("hypno_lactated_recently", False):
            return "She needs a break"
        if (0.1*(float(Person.rank_tits(person.tits)+person.lactation_sources))) * (person.arousal*1.0/person.max_arousal) > 1:
            return "Already at maximum"
        return True

    def train_hypno_lactation_requirement(person):
        if person.has_role(hypno_lactation_role):
            return False
        if person.suggestibility < 50:
            return "Requires 50% Suggestibility"
        if pregnant_tits_requirement(person):
                return True
        if not person.knows_pregnant:
            return "Requires Pregnancy"
        return "Requires Tit Growth"

    def get_hypno_lactation_role_lactation_actions():
        trigger_hypno_lactation_action = Action("Trigger lactation", trigger_hypno_lactation_requirement, "trigger_hypno_lactation", menu_tooltip = "You've implanted a trigger word. You can make her lactate whenever you want.")
        change_hypno_lactation_word_action = Action("Change lactation command word", trigger_hypno_lactation_requirement, "change_hypno_lactation_word_label", menu_tooltip = "Change the command word used to trigger lactation.")
        return [trigger_hypno_lactation_action, change_hypno_lactation_word_action]

    hypno_lactation_role = Role("Hypno Lactation", actions = get_hypno_lactation_role_lactation_actions(), hidden = True, on_turn = hypno_lactation_on_turn)

    hypno_lactation_trainable = Trainable("hypno_lactation_train", "train_hypno_lactation", "Trigger Word Lactations", base_cost = 1000, unlocked_function = train_hypno_lactation_requirement)
    special_trainables.append(hypno_lactation_trainable)

label trigger_hypno_lactation(the_person):
    $ scene_manager = Scene()
    mc.name "Hey [the_person.title]..."
    the_person "Yeah?"
    $ the_person.event_triggers_dict["hypno_lactated_recently"] = True
    $ the_word = the_person.event_triggers_dict.get("trigger_hypno_word","Lactate")
    $ the_word.capitalize()
    mc.name "[the_word]."
    $ the_person.lactation_sources += 1
    if (0.1*(float(Person.rank_tits(the_person.tits)+the_person.lactation_sources))) * (the_person.arousal_perc/100.00) < 0.50:
        if the_person.arousal < 30:
            $ the_person.change_arousal(30)
        $ scene_manager.add_actor(the_person)
        if the_person.tits_visible:
            "[the_person.title] doesn't have any idea what you did, but you can see her milk beginning to leak out of her exposed tits."
        elif the_person.wearing_bra:
            "[the_person.title] doesn't have any idea what you did, but you can see her milk beginning to leak through her bra."
        else:
            "[the_person.title] doesn't have any idea what you did, but you can see her milk beginning to soak into her clothes."
    elif (0.1*(float(Person.rank_tits(the_person.tits)+the_person.lactation_sources))) * (the_person.arousal_perc/100.00) < 1:
        if the_person.arousal < 20:
            $ the_person.change_arousal(30)
        else:
            $ the_person.change_arousal(10)
        $ scene_manager.update_actor(the_person, emotion = "happy")
        if the_person.tits_visible:
            "[the_person.title] doesn't quite seem to realise what you did, but you can see her milk clearly dripping out of her exposed tits."
        elif the_person.wearing_bra:
            "[the_person.title] doesn't quite seem to realise what you did, but you can see her milk clearly dripping through her bra."
        else:
            "[the_person.title] doesn't quite seem to realise what you did, but you can see her milk clearly soaking into her clothes."
        "She may not be entirely aware, but you see her smile widen as the physical reaction links into her emotional connection with breastfeeding."
    else:
        if the_person.arousal_perc < 10:
            $ the_person.change_arousal(30)
        else:
            $ the_person.change_arousal(20)
        $ scene_manager.update_actor(the_person, emotion = "orgasm")
        if the_person.tits_visible:
            "[the_person.title] gasps a bit and trembles as her milk lets down and you can see her milk flowing freely from her exposed tits."
        elif the_person.wearing_bra:
            "[the_person.title] gasps a bit and trembles as her milk lets down and you can see her milk flowing freely through her bra to drip down her abdomen."
        else:
            "[the_person.title] gasps a bit and trembles as her milk lets down and you can see her milk soaking right through her clothes."
        if the_person.arousal_perc >= 100:
            $ the_person.have_orgasm()
    "You are lost in the sight of her for a moment before you remember she asked you a question."
    mc.name "Oh, I was just saying how nice you look today."
    $ scene_manager.update_actor(the_person, emotion = "happy")
    the_person "Thanks [the_person.mc_title]!"
    $ del the_word
    $ scene_manager.clear_scene()
    return

label train_hypno_lactation(the_person):
    $ scene_manager = Scene()
    $ scene_manager.add_actor(the_person)
    mc.name "There's something I want to talk to you about."
    "[the_person.possessive_title!c] nods and listens attentively."
    if the_person.knows_pregnant:
        if the_person.has_child_with_mc:
            mc.name "Since you are pregnant, I was wondering if you are still breastfeeding as well."
            the_person "Yes I am, once this new baby is born I'll be breastfeeding both at the same time."
            mc.name "Won't that be kind of demanding?"
            the_person "It can be, but lots of people do it. Why do you ask?"
        else:
            if the_person.kids > 0:
                mc.name "Since you are pregnant, I was wondering if your milk had started to come in."
                the_person "Not yet, but I've done this before and it should be here when I need it."
            else:
                mc.name "Since you are pregnant, I was wondering if your milk had started to come in."
                the_person "Not yet, according to what I've read it should get here right around my due date."
                mc.name "I've heard that sometimes it can be hard to get started, and even after that keeping a steady supply can be difficult."
                the_person "It can be challenging, but I'm sure I can handle it. Why do you ask?"
                mc.name "I'd like to ensure you have enough milk for your new baby, and was researching methods to increase production."
                mc.name "Worst case scenario, you would be able to save up some extra by pumping so that if you want to take a mini vacation or have a couple drinks you don't need to worry about your milk."
                the_person "That does sound pretty good, what did you have in mind?"
    else:
        mc.name "You're breastfeeding right now, aren't you?"
        the_person "Yes."
        mc.name "I've heard that can be challenging, you never get to take even a short trip and you can't drink."
        the_person "Well not unless I've pumped enough to save up a bit of an extra supply."
        mc.name "Right, still it must be tough."
        the_person "It is a sacrifice, but one I'm happy to make."
        mc.name "Even so, I'd like to try and help you out."
        the_person "What did you have in mind?"
    if the_person.lactation_sources > 0:
        mc.name "I'd like to try and induce some extra lactation by trying to get you to let down on command."
    else:
        mc.name "I'd like to try and induce lactation on command."
    if mc.location.person_count > 1:
        the_person "Right here? There are people around..."
        mc.name "Forget about them. It is a perfectly natural part of motherhood."
    elif the_person.love < 40:
        the_person "Right in front of you? I don't know..."
        mc.name "Forget about me, this is about what you need. Think of your baby."
    else:
        the_person "Right now? I don't know..."
        mc.name "Why not right now? Come on, you know you need this."
    "She thinks about it for a moment. Her trance makes the outcome all but guaranteed."
    the_person "Okay, I'll do it. How should we start?"
    mc.name "According to my research we need to get you to relax first so lets start with a massage."
    if the_person.sluttiness_tier == 0:
        the_person "That sounds nice."
    elif the_person.sluttiness_tier == 1:
        the_person "That sounds nice, let me move some of these clothes out of the way for you."
        if the_person.wearing_bra:
            while the_person.outfit.bra_covered:
                $ generalised_strip_description(the_person, the_person.outfit.remove_random_upper(top_layer_first = True, do_not_remove = True))
        elif not the_person.tits_visible:
            while not the_person.tits_visible:
                $ generalised_strip_description(the_person, the_person.outfit.remove_random_upper(top_layer_first = True, do_not_remove = True))
        $ generalised_strip_description(the_person, the_person.outfit.get_half_off_to_tits_list(), half_off_instead = True)
    elif the_person.sluttiness_tier == 2:
        the_person "That sounds nice, let me move these clothes out of the way for you."
        $ generalised_strip_description(the_person, the_person.outfit.get_tit_strip_list())
    else:
        the_person "That sounds nice, let me get ready for you."
        $ generalised_strip_description(the_person, the_person.outfit.get_full_strip_list())
    $ the_person.update_outfit_taboos()
    $ scene_manager.update_actor(the_person, position = "walking_away")
    "You start at [the_person.title]'s shoulders, working out some basic tension."
    "The feeling of her tension draining away is pretty empowering as you work down her back."
    if the_person.sluttiness_tier > 1:
        mc.name "This is a good start, but to focus on the right body parts I'm going to need to work on your front too."
        the_person "Oh, of course."
        "You step around [the_person.possessive_title] taking in the lovely view of her chest."
        $ scene_manager.update_actor(the_person)
        mc.name "I'm going to simulate milking both of your breasts at the same time."
        the_person "That seems like the most efficient thing to do!"
        $ mc.change_locked_clarity(20)
        if the_person.lactation_sources > 0:
            "You put your hands on her engorged mammaries. The skin is hot and feels like it's pulled tight, her tits are so full of milk."
            the_person "Aahhh... I mean... it feels really good too..."
            "You start to squeeze them in rhythm. First one, then the other. It takes several seconds, but soon milk starts to dribble out of the tip."
        else:
            "You put your hands on her mammaries. The skin is hot but there doesn't seem to be much going on yet."
            the_person "Aahhh... I mean... it feels really good too..."
            "You start to squeeze them in rhythm. First one, then the other. It doesn't look like anything is happening, but you feel some slight changes."
        if the_person.sluttiness_tier > 2:
            the_person "I think it's working, but do you think maybe using your hands is a bit unnatural."
            mc.name "What do you mean?"
            the_person "Well babies suckle, and even if I pump there is suction there, I think just squeezing isn't going to be enough."
            mc.name "That sounds like a perfect next step."
            $ scene_manager.update_actor(the_person, position = "sitting")
            the_person "Come here and sit next to me."
            if the_person.lactation_sources > 0:
                "You sit down next to [the_person.title] in an awkward move, she gently lowers you on to her lap. You look up at her full, milk laden tits hungrily."
                "You bring your lips to her nipple and begin to suckle. Tiny drips begin to escape it, giving you a taste of milk."
            else:
                "You sit down next to [the_person.title] in an awkward move, she gently lowers you on to her lap. You look up at her tits hungrily."
                "You bring your lips to her nipple and begin to suckle. It takes several seconds, but soon milk starts to dribble out of the tip."
            $ the_person.change_arousal(10)
            $ mc.change_locked_clarity(30)
            "She takes one of your hands and puts it on her other breast."
            the_person "Here... squeeze this side too..."
    "It is clear that the relaxation from the massage is turning into pressure of another kind."
    $ the_word = renpy.input("Pick her trigger word. (Default: 'Lactate')", default='Lactate')
    while not the_word:
        $ the_word = renpy.input("Pick her trigger word. (Default: 'Lactate')", default='Lactate')
    $ the_person.event_triggers_dict["trigger_hypno_word"] = the_word.strip()
    "You lean very close to [the_person.possessive_title] and whisper \"[the_word]\" into her ear."
    mc.name "Focus on that as you let go. All that pressure you're feeling..."
    mc.name "Focus it all onto that one word."
    "You guide her closer and closer to her lactation, whispering reminders in her ear that she can't let down until she hears her trigger word."
    "You lean close and whisper right into her ear."
    mc.name "[the_word]."
    $ the_person.lactation_sources += the_person.sluttiness_tier
    $ the_person.event_triggers_dict["hypno_lactation_count"] = the_person.event_triggers_dict.get("hypno_lactation_count", 0) + the_person.sluttiness_tier
    $ the_person.event_triggers_dict["hypno_lactated_recently"] = True
    $ mc.change_locked_clarity(30)
    if the_person.sluttiness_tier <3:
        $ scene_manager.update_actor(the_person, emotion = "orgasm")
        $ scene_manager.update_actor(the_person, position = "sitting")
        if the_person.lactation_sources > 2:
            "[the_person.title] gives a moan, and the floodgates open as her breasts let down. Milk is now coming out in a steady stream from both tits, and every time you squeeze it spurts out forcefully."
            the_person "Oh god... that feels so good..."
            if the_person.arousal < 10:
                $ the_person.change_arousal(30)
            else:
                $ the_person.change_arousal(20)
            $ mc.change_locked_clarity(30)
            "[the_person.possessive_title!c] is panting as her tits rapidly spray milk in front of her. She is expressing an impressive amount."
        else:
            "[the_person.title] gives a moan as she lets down. Now with each squeeze, her nipples emit a short but steady squirt."
            the_person "Mmm, that feels so good..."
            if the_person.arousal < 20:
                $ the_person.change_arousal(30)
            else:
                $ the_person.change_arousal(10)
            $ mc.change_locked_clarity(10)
        $ the_person.change_stats(happiness = 5, slut = 1, max_slut = 50)
    else:
        $ scene_manager.update_actor(the_person, position = "sitting", emotion = "orgasm")
        if the_person.lactation_sources > 2:
            "Suddenly, the milk begins to flow much more rapidly as [the_person.possessive_title] starts to let down. Milk is pouring into your mouth at an alarming rate."
            "You quickly swallow gulp after gulp, just keeping up with her production. Her milk is smooth, sweet, and creamy."
            if the_person.arousal < 5:
                $ the_person.change_arousal(30)
            else:
                $ the_person.change_arousal(25)
            $ mc.arousal += 15
            $ mc.change_locked_clarity(30)
            the_person "Oh my god, yes that's it."
            "Her hand is running through your hair. It is a very intimate encounter to suckle from her like this."
        else:
            "[the_person.possessive_title!c]'s milk starts to flow more freely as she lets down. You are able to spend a few seconds at a time sucking, swallow, then repeat."
            "Her milk is smooth, sweet, and creamy."
            if the_person.arousal < 15:
                $ the_person.change_arousal(30)
            else:
                $ the_person.change_arousal(15)
            $ mc.arousal += 10
            $ mc.change_locked_clarity(10)
            the_person "Mmm, that feels really nice..."
    $ the_person.change_love(2)
    "She whispers and coos into your ear for a long moment, but little by little her lactation subsides."
    "When she is in control of herself again she stands under her own power and looks at you, a dumb smile spreading across her face."
    the_person "That was so intense... Do it again."
    $ scene_manager.update_actor(the_person)
    "You spend some more time with [the_person.possessive_title], reinforcing the strength of her trigger word."
    "When you're finished you feel confident you can use it to make her lactate on command."
    $ the_person.add_role(hypno_lactation_role)
    $ del the_word
    $ scene_manager.clear_scene()
    return True

label change_hypno_lactation_word_label(the_person):
    $ old_word = renpy.input("Pick her trigger word. (Default: '[old_word]')", default = old_word)
    while not new_word:
        $ new_word = renpy.input("Pick her trigger word. (Default: '[old_word]')", default = old_word)
    $ the_person.event_triggers_dict["trigger_hypno_word"] = new_word.strip()
    $ del new_word
    $ del old_word
    return

license: unknown


code:
init 2 python:
    change_hypno_orgasm_word_action = Action("Change orgasm command word", hypno_trigger_orgasm_requirement, "change_hypno_orgasm_word_label",
        menu_tooltip = "Change the command word used to trigger orgasms.")

init 16 python:
    add_label_hijack("normal_start", "activate_hypno_orgasm_role_enhancement")
    add_label_hijack("after_load", "activate_hypno_orgasm_role_enhancement")

label change_hypno_orgasm_word_label(the_person):
    $ scene_manager = Scene()
    mc.name "There's something I want to talk to you about."
    "[the_person.possessive_title!c] nods and listens attentively."
    mc.name "I think we should change what we use to trigger your orgasms."
    the_person "Hmm? What should I focus on now?"
    $ old_word = the_person.event_triggers_dict.get("hypno_trigger_word", "Cum")
    $ new_word = renpy.input("Pick her new trigger word. (Current: '[old_word]')", default = old_word)
    while not new_word:
        $ new_word = renpy.input("Pick her trigger word. (Current: '[old_word]')", default = old_word)
    $ the_person.event_triggers_dict["hypno_trigger_word"] = new_word.strip()
    if " " in the_person.event_triggers_dict["hypno_trigger_word"]:
        mc.name "Your new phrase should be [new_word]."
    else:
        mc.name "Your new word should be [new_word]."
    the_person "Does that mean we get to train again?"
    mc.name "We do, but it should be quicker than last time."
    "You lean close and whisper right into her ear."
    mc.name "[new_word]. [old_word]. [new_word]."
    $ scene_manager.add_actor(the_person, emotion = "orgasm")
    $ mc.change_locked_clarity(30)
    "The results are immediate. [the_person.possessive_title] spasms, bucking her hips and gasping for breath."
    the_person "Oh god! Ah! Ah!"
    "Her orgasm is so intense that her knees buckle and she starts to collapse to the ground."
    menu:
        "Catch her":
            "You slide an arm around [the_person.title] and hold her up as she cums her brains out. She clings to you on instinct with her free hand."
            "Meanwhile, her other hand doesn't stop pumping in and out of her climaxing cunt."
            $ the_person.have_orgasm()
            $ the_person.change_love(2)
            "She gasps and moans into your ear for a long moment, but little by little her orgasm subsides."
            "When she is in control of herself again she stands under her own power and looks at you, a dumb smile spreading across her face."
        "Let her fall":
            "You step back and let her climax run its course."
            $ scene_manager.update_actor(the_person, position = "doggy", emotion = "orgasm")
            "[the_person.title] falls to the ground, barely catching herself at the last minute with her free hand."
            "She ends up face down, hips bucking with each new climactic spasm. Her thighs twitch in sync, all while she continues to finger herself."
            $ the_person.have_orgasm()
            $ the_person.change_slut(2)
            "She moans and writhes on the floor for a long moment, but little by little her orgasm subsides and she gains control of herself again."
            $ scene_manager.update_actor(the_person, position = "missionary", emotion = "happy")
            "[the_person.possessive_title!c] rolls over and looks up at you, a dumb smile spreading across her face."
    the_person "That was so intense... Do it again."
    $ scene_manager.update_actor(the_person)
    if " " in the_person.event_triggers_dict["hypno_trigger_word"]:
        "You spend some more time with [the_person.possessive_title], reinforcing the strength of her trigger phrase."
        "When you're finished you feel confident you can use it to make her cum on command with the new phrase."
    else:
        "You spend some more time with [the_person.possessive_title], reinforcing the strength of her trigger word."
        "When you're finished you feel confident you can use it to make her cum on command with the new word."
    $ del new_word
    $ del old_word
    $ scene_manager.clear_scene()
    return

label activate_hypno_orgasm_role_enhancement(stack):
    python:
        hypno_orgasm_role.add_action(change_hypno_orgasm_word_action)
        execute_hijack_call(stack)
    return


code:
init 3 python:
    def social_notification_requirement():
        if len(build_insta_scroll()) + len(build_dik_scroll()) + len(build_only_scroll()) > 6:
            return True
        return False

    def build_insta_scroll():
        temp_list = []
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("insta_known", False):
                if person.event_triggers_dict.get("insta_generate_pic", False):
                    temp_list.append(person)
        return temp_list

    def build_dik_scroll():
        temp_list = []
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("dikdok_known", False):
                if person.event_triggers_dict.get("dikdok_generate_video", False):
                    temp_list.append(person)
        return temp_list

    def build_only_scroll():
        temp_list = []
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("onlyfans_known", False):
                if person.event_triggers_dict.get("onlyfans_subscription_valid_until", 0) >= day:
                    if not person.event_triggers_dict.get("onlyfans_visited_today", True):
                        temp_list.append(person)
        return temp_list

    social_notification_action = ActionMod("Social Notification",social_notification_requirement,"social_notification_action_label",
        menu_tooltip = "Your phone lights up with notifications from social media", category="Misc", is_crisis = True)

    def insta_scroll_requirement():
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("insta_known", False):
                if person.event_triggers_dict.get("insta_generate_pic", False):
                    return True
        return "Wait for tomorrow"

    def dik_scroll_requirement():
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("dikdok_known", False):
                if person.event_triggers_dict.get("dikdok_generate_video", False):
                    return True
        return "Wait for tomorrow"

    def only_scroll_requirement():
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("onlyfans_known", False):
                if not person.event_triggers_dict.get("onlyfans_visited_today", True):
                    if person.event_triggers_dict.get("onlyfans_subscription_valid_until", 0) >= day:
                        return True
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("onlyfans_known", False):
                if person.event_triggers_dict.get("onlyfans_subscription_valid_until", -1) < day:
                    return "Renew your subscriptions"
        return "Wait for tomorrow"

    def onlyfans_subscribe_requirement():
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("onlyfans_known", False):
                if person.event_triggers_dict.get("onlyfans_subscription_valid_until", -1) < day:
                    return True
        return "All subscriptions active"

    def onlyfans_renew_subscriptions(days, renew = False):
        count = 0
        for person in known_people_in_the_game():
            if person.event_triggers_dict.get("onlyfans_known", False):
                if person.event_triggers_dict.get("onlyfans_subscription_valid_until", -1) < day:
                    if renew:
                        person.event_triggers_dict["onlyfans_subscription_valid_until"] = day + days
                    else:
                        count +=1
        return count

    def build_phone_menu_social_scroll_extended(org_func):
        def phone_menu_wrapper():
            # run original function
            phone_menu = org_func()
            # run extension code
            scroll_insta_action = Action("Scroll InstaPic", insta_scroll_requirement, "scroll_insta")
            scroll_dikdok_action = Action("Scroll Dikdok", dik_scroll_requirement, "scroll_dikdok")
            scroll_onlyfans_action = Action("Scroll OnlyFanatics", only_scroll_requirement, "scroll_onlyfans")
            onlyfans_subscribe_action = Action ("Renew OnlyFanatics Subs", onlyfans_subscribe_requirement, "onlyfans_subscribe_label")
            phone_menu[1].insert(2, scroll_insta_action)
            phone_menu[1].insert(4, scroll_dikdok_action)
            phone_menu[1].insert(6, scroll_onlyfans_action)
            phone_menu[1].insert(7, onlyfans_subscribe_action)
            return phone_menu
        return phone_menu_wrapper

    if "build_phone_menu" in globals():
        build_phone_menu = build_phone_menu_social_scroll_extended(build_phone_menu)

label scroll_insta():
    $ insta_scroll = build_insta_scroll()
    $ insta_count = len(build_insta_scroll())-1
    while insta_count > -1:
        call view_insta_enhanced(insta_scroll[insta_count]) from _call_view_insta_scroll
        menu:
            "Keep scrolling" if insta_count > 0:
                $ insta_count -= 1
            "Stop":
                $ insta_count = -1
    return

label view_insta_enhanced(the_person):
    $ scene_manager = Scene()
    $ posted_today = True
    if the_person.event_triggers_dict.get("insta_special_request_outfit", None):
        $ skimpy_outfit = the_person.event_triggers_dict.get("insta_special_request_outfit", insta_wardrobe.pick_random_outfit())
    else:
        $ skimpy_outfit = insta_wardrobe.pick_random_outfit()
    if the_person.event_triggers_dict.get("insta_new_boobs_brag", None) is not None:
        $ the_person.event_triggers_dict["insta_new_boobs_brag"] = None
        $ rand_num = renpy.random.randint(0,2)
        if rand_num == 0:
            $ scene_manager.add_actor(the_person, skimpy_outfit)
        else:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "kneeling1")
        $ mc.change_locked_clarity(15)
        the_person "Went to the doc and got some upgrades! Don't they look great?!" (what_style = "text_message_style")
        $ the_person.apply_outfit() # Reset them to their normal daily wear.
        $ del skimpy_outfit
    elif the_person.event_triggers_dict.get("insta_special_request_outfit", None):
        $ ran_num = renpy.random.randint(0,3)
        if ran_num == 0:
            $ scene_manager.update_actor(the_person, skimpy_outfit)
        elif ran_num == 1:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "kneeling1")
        elif ran_num == 2:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "back_peek")
        $ mc.change_locked_clarity(10)
        the_person "Wearing something special today: a design sent in by a fan!" (what_style = "text_message_style")
        $ the_person.event_triggers_dict["insta_special_request_outfit"] = None
    elif the_person.effective_sluttiness() + the_person.get_opinion_score(["showing her ass", "showing her tits"]) * 5 > 20: #TODO: Decide what slut_requirement should be.
        $ ran_num = renpy.random.randint(0,3)
        if ran_num == 0:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "stand3")
            $ mc.change_locked_clarity(5)
            the_person "Thought this outfit looked sexy. What do you think?" (what_style = "text_message_style")
        elif ran_num == 1:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "kneeling1")
            $ mc.change_locked_clarity(10)
            the_person "Hey everyone, what do you think of this pose? I think it makes my tits look great!" (what_style = "text_message_style")
        elif ran_num == 2:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "back_peek")
            $ mc.change_locked_clarity(5)
            the_person "Ass was looking great, just had to take a pic!" (what_style = "text_message_style")
        elif ran_num == 3:
            $ scene_manager.update_actor(the_person, skimpy_outfit, position = "kneeling1")
            $ mc.change_locked_clarity(10)
            the_person "Do I look good down on my knees?" (what_style = "text_message_style")
        if the_person.has_role(dikdok_role) and the_person.event_triggers_dict.get("dikdok_generate_video", False):
            the_person "If you liked that, come see me getting into trouble on DikDok! Hurry, I might get banned soon!" (what_style = "text_message_style")
            $ the_person.event_triggers_dict["dikdok_known"] = True
        elif the_person.has_role(onlyfans_role) and the_person.event_triggers_dict.get("instafans_generate_content", False):
            the_person "If you like that, subscribe to my OnlyFanatics and see soooo much more!" (what_style = "text_message_style")
            $ the_person.event_triggers_dict["onlyfans_known"] = True
    elif the_person.is_wearing_uniform and not (the_person.vagina_visible or the_person.tits_visible):
        $ ran_num = renpy.random.randint(0,1)
        if ran_num == 0:
            $ mc.change_locked_clarity(5)
            $ scene_manager.update_actor(the_person)
            the_person "Getting dressed for work. I make this uniform work!" (what_style = "text_message_style")
        elif ran_num == 1:
            $ mc.change_locked_clarity(10)
            $ scene_manager.update_actor(the_person, position = "back_peek")
            the_person "I think my boss makes me wear this just because it makes my butt look good. At least he's right!" (what_style = "text_message_style")
    else:
        $ ran_num = renpy.random.randint(0,1)
        if ran_num == 0:
            $ scene_manager.update_actor(the_person)
            the_person "Good morning everyone!"
        elif ran_num == 1:
            $ scene_manager.update_actor(the_person, position = "back_peek")
            the_person "About to head out the door. I've got a full day ahead of me!"
    $ del skimpy_outfit
    $ scene_manager.clear_scene()
    $ the_person.event_triggers_dict["insta_generate_pic"] = False
    return

label scroll_dikdok():
    $ dik_scroll = build_dik_scroll()
    $ dik_count = len(build_dik_scroll())-1
    while dik_count > -1:
        call view_dikdok(dik_scroll[dik_count]) from _call_view_dik_scroll
        menu:
            "Keep scrolling" if dik_count > 0:
                $ dik_count -= 1
            "Stop":
                $ dik_count = -1
    return

label scroll_onlyfans():
    $ only_scroll = build_only_scroll()
    $ only_count = len(build_only_scroll())-1
    while only_count > -1:
        call view_onlyfans(only_scroll[only_count]) from _call_view_onlyfans_scroll
        menu:
            "Keep scrolling" if only_count > 0:
                $ only_count -= 1
            "Stop":
                $ only_count = -1
    return

label social_notification_action_label():
    $ scene_manager = Scene()
    $ insta_count = len(build_insta_scroll())-1
    $ dik_count = len(build_dik_scroll())-1
    $ only_count = len(build_only_scroll())-1
    $ insta_scroll = build_insta_scroll()
    $ dik_scroll = build_dik_scroll()
    $ only_scroll = build_only_scroll()
    if time_of_day == 0:
        "As you reach for your phone where it was charging overnight you see the screen light up."
    elif time_of_day == 4:
        "After you plug your phone in for the night you see it light up with some kind of message."
    elif mc.is_at_work:
        "As you are working you get what turns out to be a phantom vibration from the phone in your pocket, but while checking it you see something on the screen."
    elif mc.is_home:
        "As you walk back into your room after a quick bathroom break you see your phone screen light up."
    else:
        "You pull out your phone to pass a bit of downtime and notice something on the screen."
    if insta_count >= dik_count:
        if insta_count >= only_count:
            $ choice = "InstaPic"
            $ number = insta_count+1
        else:
            $ choice = "OnlyFanatics"
            $ number = only_count+1
    elif dik_count >= only_count:
        $ choice = "DikDok"
        $ number = dik_count+1
    else:
        $ choice = "OnlyFanatics"
        $ number = only_count+1
    "It looks like you just got a notification from [choice]."
    "As a matter of fact, you've racked up [number] notifications from [choice] since you last checked it out."
    $ number = mc.location.person_count
    if number < 1:
        "You're alone, so you don't have to worry about what comes up on your screen."
    elif number < 2:
        "There is only one person around, as long as you are careful she shouldn't see or hear anything on your phone."
    else:
        "There are a few people around, but you can angle your phone so they can't see anything."
    menu:
        "Take a look":
            $ done = False
            "You've got some time so you scan the room again and settle into a comfortable position."
            "You attach your earbuds just to be safe and unlock your phone."
        "Ignore them":
            $ done = True
            "You don't have time for this, you get back to what you were doing."
    while not done:
        $ insta_count = len(build_insta_scroll())-1
        $ dik_count = len(build_dik_scroll())-1
        $ only_count = len(build_only_scroll())-1
        $ insta_scroll = build_insta_scroll()
        $ dik_scroll = build_dik_scroll()
        $ only_scroll = build_only_scroll()
        if choice == "InstaPic":
            while insta_count > -1:
                call view_insta(insta_scroll[insta_count]) from _call_view_insta_event
                menu:
                    "Keep scrolling":
                        $ insta_count -= 1
                    "Stop":
                        $ insta_count = -1
        elif choice == "DikDok":
            while dik_count > -1:
                call view_dikdok(dik_scroll[dik_count]) from _call_view_dik_event
                menu:
                    "Keep scrolling":
                        $ dik_count -= 1
                    "Stop":
                        $ dik_count = -1
        elif choice == "OnlyFanatics":
            while only_count > -1:
                call view_onlyfans(only_scroll[only_count]) from _call_view_onlyfans_event
                menu:
                    "Keep scrolling":
                        $ only_count -= 1
                    "Stop":
                        $ only_count = -1
        $ insta_count = len(build_insta_scroll())
        $ dik_count = len(build_dik_scroll())
        $ only_count = len(build_only_scroll())
        "You close out the [choice] app, but while holding your phone you are reminded of the other social media apps you have sitting unused."
        "Would you like to check one of them?"
        menu:
            "Check InstaPic [insta_count]" if not choice == "InstaPic" and insta_count > 0:
                $ choice = "InstaPic"
            "Check DikDok [dik_count]" if not choice == "DikDok" and dik_count > 0:
                $ choice = "DikDok"
            "Check OnlyFanatics [only_count]" if not choice == "OnlyFanatics" and only_count > 0:
                $ choice = "OnlyFanatics"
            "No":
                $ done = True
    $ scene_manager.clear_scene()
    $ del insta_count
    $ del dik_count
    $ del only_count
    $ del insta_scroll
    $ del dik_scroll
    $ del only_scroll
    $ del done
    $ del choice
    $ del number
    return

label onlyfans_subscribe_label():
    $ days = -1
    $ count = onlyfans_renew_subscriptions(0, renew = False)
    $ total_count = len(build_only_scroll()) + count
    $ day_cost = 5*count
    $ week_cost = 20*count
    $ month_cost = 60*count
    "You follow [total_count] people on OnlyFanatics but [count] of your subscriptions have lapsed so you can't see their content."
    "Would you like to renew all of your inactive subscriptions?"
    menu:
        "Subscribe for a day\n{color=#ff0000}{size=18}Costs: $[day_cost]{/size}{/color}" if mc.business.has_funds(day_cost):
            $ mc.business.change_funds(-day_cost)
            $ days = 0
        "Subscribe for a day\n{color=#ff0000}{size=18}Costs: $[day_cost]{/size}{/color} (disabled)" if not mc.business.has_funds(day_cost):
            pass
        "Subscribe for a week\n{color=#ff0000}{size=18}Costs: $[week_cost]{/size}{/color}" if mc.business.has_funds(week_cost):
            $ mc.business.change_funds(-week_cost)
            $ days = 7
        "Subscribe for a week\n{color=#ff0000}{size=18}Costs: $[week_cost]{/size}{/color} (disabled)" if not mc.business.has_funds(week_cost):
            pass
        "Subscribe for a month\n{color=#ff0000}{size=18}Costs: $[month_cost]{/size}{/color}" if mc.business.has_funds(month_cost):
            $ mc.business.change_funds(-month_cost)
            $ days = 30
        "Subscribe for a month\n{color=#ff0000}{size=18}Costs: $[month_cost]{/size}{/color} (disabled)" if not mc.business.has_funds(month_cost):
            pass
        "Not now":
            pass
    if days > -1:
        $ onlyfans_renew_subscriptions(days, renew = True)
        $ only_count = len(build_only_scroll())-1
        $ only_scroll = build_only_scroll()
        "With your subscriptions renewed you have [only_count] unseen posts waiting for you. Would you like to watch them now?"
        menu:
            "Yes":
                while only_count > -1:
                    call view_onlyfans(only_scroll[only_count]) from _call_view_onlyfans_sebscribe
                    menu:
                        "Keep scrolling":
                            $ only_count -= 1
                        "Stop":
                            $ only_count = -1
            "No":
                pass
        $ del only_count
        $ del only_scroll
    $ del days
    $ del count
    $ del total_count
    $ del day_cost
    $ del week_cost
    $ del month_cost
    return


code:
init -1 python:
    def baby_fever_get_people():
        temp_list = []
        final_list = []
        for person in known_people_in_the_game():
            if person not in [lily, mom, cousin, aunt]:
                if person.location == mc.location:
                    if not person.event_triggers_dict.get("preg_knows", False): # not pregnant
                        if person.event_triggers_dict.get("last_birth", 0) < 1: # no baby
                            if person.event_triggers_dict.get("baby_fever_chat", -1) < day -14: # no recent talk
                                if pregnant_people(person):
                                    temp_list.append(person)
        if temp_list:
            while len(final_list) < 1:
                target = mc.business.event_triggers_dict.get("baby_fever", -4)
                for person in temp_list:
                    if person.opinion.creampies + person.opinion.bareback_sex == target:
                        final_list.append(person)
                if not final_list:
                    if target == 4:
                        mc.business.event_triggers_dict["baby_fever"] = -4
                    else:
                        mc.business.event_triggers_dict["baby_fever"] = target + 1
        return final_list

    def crisis_baby_fever_requirement():
        if mc.business.event_triggers_dict.get("baby_fever_chat", -1) < day - 5:
            if baby_fever_get_people():
                return True
        return False

init 3 python:
    crisis_baby_fever_action = ActionMod("Baby Fever", crisis_baby_fever_requirement, "crisis_baby_fever_label",
        menu_tooltip = "A person is jealous of an expectant mother.", category = "Misc", is_crisis = True, is_morning_crisis = False, priority = 5)

label crisis_baby_fever_label():
    $ scene_manager = Scene()
    $ mc.business.event_triggers_dict["baby_fever_chat"] = day
    $ the_person = get_random_from_list(baby_fever_get_people())
    $ the_person.event_triggers_dict["baby_fever_chat"] = day
    $ influence = the_person.suggestibility
    if pregnant_family(the_person):
        $ the_mother = get_random_from_list(pregnant_family(the_person))
    else:
        $ the_mother = get_random_from_list(pregnant_people(the_person))
    $ so_title = SO_relationship_to_title(the_person.relationship)
    $ opinion = the_person.opinion.creampies + the_person.opinion.bareback_sex
    $ relationship = town_relationships.get_relationship(the_mother, the_person).type_a
    $ scene_manager.add_actor(the_person, emotion="happy")
    "You're not really doing anything at the moment so you take a look around the room."
    "It seems like you aren't the only one bored, [the_person.title] is also sort of staring off into space but seems to notice your eyes pass over her."
    "[the_person.possessive_title!c] walks up to you and strikes up a conversation. After a bit of small talk it is clear she has something specific on her mind."
    if the_mother not in known_people_in_the_game():
        the_person "I have sort of a weird question for you. Do you know [the_mother.fname] [the_mother.last_name]?"
        mc.name "Sorry, I can't say that I do."
        the_person "Oh, well I'll have to introduce you sometime."
        # on enter event: [planned_intro] = person req both together
    elif the_mother in [mom, lily, cousin, aunt]:
        the_person "Congratulations by the way. I've been meaning to talk to you for a while."
        mc.name "Congratulations?"
        the_person "Yeah, you know, [the_mother.fname] is going to have her baby soon."
        "What!? Is [the_mother.possessive_title] telling people you are the father of her baby?"
        mc.name "...oh right... thanks?"
        the_person "Sorry, I guess that is a little weird. Congratulations isn't exactly what I meant."
        mc.name "Umm..."
        if the_mother == mom:
            the_person "Still, you must be excited. Soon you are going to get to be a big brother all over again."
            "Oh right, your new sibling..."
        elif the_mother == lily:
            the_person "Still, you must be excited. You're going to be an uncle soon."
            "Oh right, your new niece or nephew..."
        elif the_person == cousin:
            the_person "Still, you must be excited. The new baby is going to be your second cousin right?"
            mc.name "Actually they will be my first cousin once removed."
            the_person "Oh, I can never keep them straight. I just call everyone in my extended family cousins."
        else:
            the_person "Still, you must be excited. Another cousin for you and your sister."
            "Oh right, your new cousin..."
        "It seems like your family tree is going to be a bit more tangled than most going forward."
        mc.name "Right. Thank you! I'm really looking forward to the new baby."
    elif the_mother.type == "story":
        the_person "I've been thinking about [the_mother.fname] recently."
    else:
        the_person "You know [the_mother.fname] right?"
    the_person "We were talking the other day and I couldn't help but notice her body."
    mc.name "Really? I didn't know you were interested in girls [the_person.title]."
    the_person "What, no that isn't what I mean. The way her body is changing."
    if the_mother.event_triggers_dict.get("preg_transform_day", 0) > day:
        the_person "It is subtle, but I can tell something is going on. I'm pretty sure she is pregnant."
        mc.name "Really? I guess maybe I need to spend more time staring at her boobs."
    else:
        the_person "It is pretty clear that something is different. You know she is pregnant right?"
        mc.name "Oh that, of course. I mean I'd have to be blind to miss the signs by now."
    if the_person.has_role(girlfriend_role):
        the_person "Well, don't look too hard. I would think having me around would give you enough eye candy."
    mc.name "So, is this just idle gossip or..."
    the_person "Actually, seeing her made me think about my body a bit."
    if the_person.body_type == "thin_body":
        the_person "I've always taken pride in my body, keeping it toned and fit."
    elif the_person.body_type == "standard_body":
        the_person "I've always felt comfortable in my body without working too hard."
    else:
        the_person "I've always been a little self-conscious about my body."
    the_person "Seeing her, and the way she is growing has given me a new perspective."
    mc.name "Really?"
    if opinion < 0:
        the_person "Yeah, she is blowing up like a balloon. She looks like she can barely fit in her clothes."
        the_person "It gives me one more reason to avoid getting knocked up anytime soon."
        $ the_person.update_birth_control_knowledge()
        if the_person.on_birth_control:
            if opinion < -2:
                the_person "I mean, I'm on birth control and I still try to be careful about condoms."
            else:
                the_person "I mean, there is a reason I take birth control."
        else:
            if opinion < -2:
                the_person "I'm not taking birth control right now but I try to be careful with condoms."
            else:
                the_person "I should really get myself some new birth control pills."
        mc.name "Oh, well motherhood isn't for everyone."
        the_person "Exactly... although there is one aspect that does make me sort of jealous."
    else:
        the_person "Yeah, I mean she is certainly bigger than me, but it looks so natural and beautiful."
        the_person "There is this glow about her that I can't really define."
        $ the_person.update_birth_control_knowledge()
        if the_person.on_birth_control:
            if opinion < 2:
                the_person "My hormones are urging me to forget my birth control, but I'll probably keep using other protection."
            else:
                the_person "It sort of makes me want to stop my birth control and see what happens."
        else:
            if opinion < 2:
                the_person "My hormones have been a bit crazy without birth control stabilizing them."
            else:
                the_person "I'm glad I'm not on birth control, the idea of being filled up without protection is kind of exciting."
        mc.name "Well you would make an excellent mother if my opinion means anything."
        the_person "Thanks, plus there would be at least one other significant advantage to getting knocked up."
    the_person "Her tits are so full she has been practically spilling out of her clothes."
    the_person "There is something primal about seeing her swollen breasts. I get this unexplained desire that just fills my body."
    mc.name "I know exactly what you mean."
    the_person "Yeah, I'm sure you do."
    if Person.rank_tits(the_person.tits) > Person.rank_tits(the_mother.tits):
        the_person "Of course I still have the advantage on size, although if mine were to swell like hers they would be truly epic."
    else:
        the_person "It makes me think about getting a boob job so that I can keep up with her, or maybe even go further."
    the_person "I'd look incredible with bigger tits, don't you think [the_person.mc_title]?"
    if the_person.sluttiness > 40:
        "To drive her point home [the_person.title] reaches up and grabs her tits, pushing them up to accentuate her cleavage."
        $ mc.change_locked_clarity(10)
    mc.name "You absolutely would."
    "She seems lost in thought for a moment and you take advantage of the break in the conversation to stare at her body a bit."
    $ body_type = the_person.body_type
    $ tit_size = the_person.tits
    $ the_person.tits = Person.get_larger_tit(the_person.tits)
    $ scene_manager.update_actor(the_person, position="stand2", emotion="happy")
    "[the_person.possessive_title!c] does have amazing tits, but seeing them grow naturally as part of her pregnancy would be almost as enjoyable as the act of impregnating her."
    $ the_person.tits = Person.get_larger_tit(the_person.tits)
    $ the_person.outfit.strip_to_underwear()
    $ scene_manager.update_actor(the_person, position="stand3", emotion="happy")
    "Watching them grow day by day until she could barely contain them in her bra would certainly make your time with her more enjoyable."
    "Then her belly would start to swell with your seed, filling her up with a new life that you created together."
    $ the_person.body_type = "standard_preg_body"
    $ scene_manager.update_actor(the_person, position="stand4", emotion="happy")
    "Watching her body grow and transform in service to your needs and demands would fill you with pride."
    "Knowing her devotion and desire to serve you is such a driving force that she is willing to carry your children would certainly add to the pleasure you take in her body."
    $ scene_manager.update_actor(the_person, position="back_peek", emotion="happy")
    the_person "...[the_person.mc_title]?"
    $ the_person.tits = tit_size
    $ the_person.body_type = body_type
    if the_person.is_employee and the_person.is_at_office:
        $ the_person.wear_uniform()
    else:
        $ the_person.apply_planned_outfit()
    $ scene_manager.update_actor(the_person, emotion="sad")
    "Oh shit! Was she talking to you?"
    mc.name "Sorry, what was that?"
    the_person "I said, do you think it would be crazy for me to try and get pregnant?"
    menu:
        "Encourage motherhood":
            $ influence +=20
            mc.name "I think getting pregnant and having a baby would be great for you."
            mc.name "You would make an incredible mother, and I'd love to see your body swell through your pregnancy."
            the_person "You really think so? [the_mother.fname] looks so fulfilled, I'd love to feel that way too."
        "Discourage motherhood":
            $ influence -=20
            mc.name "Babies might be great, but you don't want to sacrifice the life you have now."
            mc.name "I don't think you have thought about all the impacts a baby would have on your life."
            the_person "You might be right, she just looks so fulfilled."
    mc.name "You don't have to decide right now. It is a big decision, you should weigh the options."
    the_person "I know, it's just..."
    if relationship == "Mother":
        the_person "Seeing my mom pregnant again really makes me think about starting a family of my own."
        the_person "I already know she is a great mom, her baby is going to be so lucky."
    elif relationship == "Daughter":
        the_person "I'm so excited about having a grandbaby, but it really makes me think about how happy I was raising [the_mother.fname]."
        if the_mother.kids < 1:
            the_person "I can't wait to see how she does as a mother."
    elif relationship == "Sister":
        the_person "I can't wait to be an aunt, but it makes me a bit jealous of my [relationship!l] too."
        if the_person.age > the_mother.age:
            the_person "I mean, I am older than her and I don't have any kids of my own."
        else:
            the_person "Although I guess I still have time. She is older than me after all."
    elif relationship == "Rival" or relationship == "Nemesis":
        the_person "I hate seeing her so happy."
        if the_mother.kids < 1:
            the_person "It kills me to admit it, but she is probably going to make a great mom."
    else:
        the_person "I love seeing her so happy."
        if the_mother.kids < 1:
            the_person "She has so much love to give, she'll make such a great mom."
    if not the_person.is_single and not the_person.is_girlfriend:
        if the_person.kids < 1:
            the_person "I've started thinking about talking to [the_person.SO_name], see if maybe we could get serious about starting a family."
        else:
            the_person "I've been thinking about talking to [the_person.SO_name] about giving me a new baby."
        menu:
            "Offer an alternative":
                mc.name "You know, if you're serious about starting a family I would be happy to help you with that."
                if the_person.love > leave_SO_love_calculation(the_person) - 10:
                    the_person "Wait, are you serious? I'd love to start a family with you. I just thought I'd have to do it with my [so_title]."
                    mc.name "Absolutely, tell him you are through and the two of us can start planning for the future."
                    call transform_affair(the_person) from _call_transform_affair_baby
                    $ the_person.change_love(10)
                    $ the_person.change_obedience(5)
                    $ scene_manager.update_actor(the_person, position = "kissing", special_modifier = "kissing")
                    "You put your arms around her waist and she kisses you immediately. When you break the kiss she's grinning ear to ear."
                    $ scene_manager.update_actor(the_person, emotion = "happy")
                    $ ex_title = so_title[:4]
                    the_person "It feels so good to not have to hide anything anymore! I'll break the news to my [ex_title]... My ex-[so_title] later today."
                else:
                    the_person "You can't be serious. I already have a [so_title]. Do you really think I'd leave him for you just to have a kid?"
                    mc.name "Sorry, I wasn't thinking clearly."
            "Raise concerns":
                if the_person.relationship == "Girlfriend":
                    mc.name "Your [so_title]? Are you sure that relationship is going to last?"
                    mc.name "I don't mean to question your choice in men, but he might not be there for you long term."
                elif the_person.relationship == "Fiancée":
                    mc.name "Your [so_title]? How long have you been engaged now?"
                    mc.name "You might want to wait until the marriage to commit to such a life-changing decision."
                else:
                    mc.name "Your [so_title]? That makes sense."
                    if the_person.kids == the_person.number_of_children_with_mc:
                        mc.name "It's actually a little surprising you haven't started one already. Does he want kids?"
                    else:
                        mc.name "I mean you already have kids with him right? Does he want more?"
                    the_person "I think so..."
                the_person "Maybe you're right, I guess I need to talk with him."
                mc.name "You should definitely do that soon."
                $ relation_ship_quarrel = Action("Girl had a fight with her SO", so_relationship_quarrel_requirement, "so_relationship_quarrel_label",
        event_duration = 2, priority = 1, trigger = "on_talk")
                $ the_person.on_talk_event_list.append(relation_ship_quarrel)
            "Show support":
                mc.name "That's great! I'm sure he would be thrilled to give you a child."
    elif the_person.is_girlfriend:
        the_person "I know we are just dating, but do you think we could talk about starting a family?"
        menu:
            "Absolutely":
                mc.name "I'd love to start a family with you [the_person.title]."
                the_person "That is so great to hear, I think we could raise incredible kids."
            "Not yet":
                mc.name "I don't think I'm ready for that. What we have is great, but it is still so new."
                the_person "Of course, no pressure. Just my hormones going crazy."
    else:
        the_person "I don't even have anyone that would be able to father the children."
        menu:
            "Offer your seed":
                mc.name "You know, if you are looking for a donor I would be happy to help you out."
                if the_person.has_taboo("vaginal_sex"):
                    "She is stunned by this statement and stares at you in disbelief for a moment."
                    the_person "You can't be serious? Is this some kind of weird way to try and get into my pants?"
                    mc.name "Hey, I was just offering. You can say no."
                elif the_person.has_taboo("condomless_sex"):
                    the_person "Trying to finally get into me bare [the_person.mc_title]?"
                    mc.name "I admit, that would be a perk I'll happily accept."
                else:
                    the_person "That could be fun, I'm assuming you mean trying naturally right?"
                    mc.name "Of course, no need to involve medical professionals if we can get there without them."
            "Start a relationship" if the_person.love > 60:
                mc.name "Actually I have a solution for that problem. I've been thinking about this for a while now."
                mc.name "I really like you and I hope you feel the same way about me."
                mc.name "I'd like to make our relationship official. What do you say?"
                $ the_person.change_happiness(15)
                $ the_person.change_love(5)
                if the_person.age > 40:
                    the_person "Oh I'm so happy to hear you say that! I was worried about our age difference, but I don't want that to stop us!"
                else:
                    the_person "Oh my god, I'm so happy! Yes, I want to be your girlfriend!"
                "She puts her arms around you and pulls you close."
                $ mc.change_locked_clarity(10)
                "She kisses you, and you kiss her back just as happily."
                $ the_person.add_role(girlfriend_role)
            "Stay silent":
                pass
    if opinion < 4:
        if the_person.on_birth_control:
            the_person "That also means going off birth control."
            menu:
                "Stop your birth control":
                    mc.name "You should stop taking it."
                    the_person "Do you really think so?"
                    mc.name "Yes, the sooner you stop taking those pills the sooner you can try getting knocked up."
                    $ manage_bc(the_person, start = False)
                "Keep taking birth control":
                    mc.name "Let's not be too hasty, this is something you should think about."
                    the_person "Yeah, I guess."
                    mc.name "You have time, be sure you really want this before you do anything rash."
        else:
            the_person "It's not like I've been working too hard to avoid having kids."
        if the_person.opinion.bareback_sex > 1:
            the_person "Plus it is always great to have a guy enter me raw."
            if the_person.opinion.creampies > 1:
                the_person "And when they finish inside me I nearly always cum."
            else:
                the_person "Although cleaning up after someone fills me up can be annoying."
            $ the_person.discover_opinion("creampies")
        else:
            the_person "Although I've always felt safer with my partner using a condom."
        $ the_person.discover_opinion("bareback sex")
        menu:
            "Try to influence her":
                $ number = -2
                if the_person.has_role(trance_role):
                    $ influence +=20
                if the_person.has_exact_role(heavy_trance_role):
                    $ influence +=20
                elif the_person.has_exact_role(very_heavy_trance_role):
                    $ influence +=40
                while number < 2:
                    if the_person.opinion.bareback_sex == number:
                        $ number = 2
                        mc.name "You know, if you really want to have amazing sex the only way to go is raw."
                        mc.name "There is nothing like feeling the pulsing heat of your partner."
                        if renpy.random.randint(1, 100) < influence:
                            the_person "I know what you mean. I've just always been too worried about the consequences."
                            $ the_person.increase_opinion_score("bareback sex")
                            the_person "Maybe I'll give it a try next time and see how it feels."
                        else:
                            the_person "Sorry, the risk just isn't worth it to me."
                    elif the_person.opinion.creampies == number:
                        $ number = 2
                        mc.name "You know, if you've never felt a man blow his load deep inside than you don't know what you're missing."
                        mc.name "The biological clock is a powerful force, and I've known girls who cum just because their body feels like it is accomplishing that most basic imperative."
                        if renpy.random.randint(1, 100) < influence:
                            the_person "That does make sense, I've just always hated the clean-up afterwards."
                            $ the_person.increase_opinion_score("creampies")
                            the_person "Maybe I'll give it a try next time and see how it feels."
                        else:
                            the_person "Sorry, the risk just isn't worth it to me."
                    else:
                        $ number +=1
            "Leave her opinions alone":
                mc.name "I guess you're right."
    elif not the_person.has_role(breeding_fetish_role):
        mc.name "You know, this conversation reminds me..."
        $ start_breeding_fetish_quest(the_person)
        call train_breeder_label(the_person) from _call_train_breeder_label_baby
        $ the_person.update_birth_control_knowledge()
    else:
        mc.name "You know, I think I have a way to make you happy."
        call breeding_fetish_bend_her_over_label(the_person) from _call_breeding_fetish_bend_her_over_label_baby
    $ scene_manager.clear_scene()
    return

label lily_baby_fever():
    $ the_person = lily
    $ the_mothers = pregnant_people(the_person)
    $ the_mother = get_random_from_list(the_mothers)
    if the_mother == mom:
        the_person "I can't believe mom is pregnant at her age."
        the_person "I guess she isn't really that old, but by the time this baby is grown up she will be."
        if mc.business.event_triggers_dict.get("family_threesome", False) == True:
            the_person "Why did you have to cum inside her so much [the_person.mc_title]?"
            the_person "I know you're the father, it's not like the two of you have been careful about using protection."
            the_person "I swear, the number of cum stains I've seen in her bedroom makes it look like you are the one who lives in there."
            the_person "Do you love her more than me?"
            the_person "I'm here for you too and my tight young pussy has got to be way more fertile than hers."
            the_person "Seeing the way she has been glowing the last few days has got me thinking."
        else:
            the_person "Who even is the father? It's not like she has been dating anyone."
            the_person "Do you think it could be someone she works with? That is the only place she really goes when she isn't home."
        the_person "There's no way that I really remember nursing from her, but sometimes when she leans over by me I get this bizarre muscle memory reaction like I want to suckle from her."
        mc.name "You know, if you wanted to ask her about her milk she might want some help."
        mc.name "She doesn't really have a use for it yet and I'm sure the pressure can get painful when they fill up too much."
        #write another morning crisis
    if the_mother == cousin:
        the_person "Have you seen how big [the_mother.fname] is getting?"
        the_person "She has always been a bit bigger than me but it is getting ridiculous as she swells up with her baby."
        the_person "And have you seen her tits? They are huge and starting to leak a bit. When I see them I get the weirdly compelling thirst."
        the_person "Do you know who the father is? Every time I try to discuss it she gets mad at me."
        the_person "TEXT"
    if the_mother == aunt:
        pass
    if the_mother == erica and "threesome" in erica_get_wakeup_options():
        pass
    if the_mother == sarah and Person.get_person_by_identifier(sarah.event_triggers_dict.get("initial_threesome_target", None)) == the_person:
        pass
    if the_person.sex_record.get(oral, 0) > 0:
        if the_person.sex_record.get("vaginal_sex", 0) == 0:
            the_person "I know we haven't gone all the way yet, but I've been thinking that maybe we could try it."
        if the_person.sex_record.get(anal, 0) > 0:
            the_person "I love it in my ass, but my pussy is ready for you too."
        else:
            if the_person.sex_record.get("condomless_sex", 0) > 0:
                the_person "You've always been careful about wearing a condom, but I think I'm ready to feel what it is like to get fucked by you bare."
            elif the_person.sex_record.get("vaginal_creampies", 0) > 0:
                the_person "You've been so careful not to cum inside me, but I'm starting to hope that you will."
                the_person "I want to know what it feels like to have your cum slosh around inside my pussy, and if you knock me up that wouldn't be so bad."
        the_person "I think I'm ready to go off birth control and have you breed me. I can't imagine a more perfect way for you to express your love than to give me a baby."
        the_person "I'm already off birth control, you'll just have to cum inside me more often so that I can be bred by you."
    return


code:
init -1 python:
    def get_person_bra_size(self):
        if not hasattr(self, "_bra_size"):
            self._bra_size = Person.rank_tits(self.tits)
        return self._bra_size

    def set_person_bra_size(self, value):
        self._bra_size = value
        return

    Person.bra_size = property(get_person_bra_size, set_person_bra_size, None, "The bra size of a person.")

    def crisis_bra_size_requirement():
        if mc.business.is_open_for_business:
            if mc.is_at_work:
                if crisis_bra_size_get_people():
                    return True
        return False

    def bra_count(person):
        count = 0
        for outfit in [x for x in person.wardrobe.outfit_sets + person.wardrobe.underwear_sets if x.wearing_bra]:
            count += 1
        if count < 1 and person.wearing_bra:
            count += 1
        return count

    def crisis_bra_size_get_people():
        temp_list = []
        for person in mc.business.employee_list:
            if not pregnant_role in person.special_role:
                if person.bra_size + 1 < Person.rank_tits(person.tits): #more than 1 size bigger
                    if bra_count(person) > 0:
                        temp_list.append(person)
                    elif person.bra_size + 2 < Person.rank_tits(person.tits):
                        temp_list.append(person)
                elif person.bra_size - 1 > Person.rank_tits(person.tits): #more than 1 size smaller
                    if bra_count(person) > 0:
                        temp_list.append(person)
            for person in temp_list:
                for serum in person.serum_effects:
                    if serum.has_trait(breast_enhancement) or serum.has_trait(breast_reduction) or serum.has_trait(lactation_hormones):
                        if person in temp_list:
                            temp_list.remove(person)
        return temp_list

init 3 python:
    crisis_bra_size_action = ActionMod("Bra Size", crisis_bra_size_requirement, "crisis_bra_size_label",
        menu_tooltip = "An employee lets you know their bras don't fit anymore.", category = "Business Crisis", is_crisis = True, is_morning_crisis = False, priority = 5)

label crisis_bra_size_label():
    $ scene_manager = Scene()
    $ the_person = get_random_from_list(crisis_bra_size_get_people())
    $ test_outfit = the_person.outfit.get_copy()
    $ the_bra = test_outfit.get_bra()
    if the_bra:
        $ test_outfit.remove_clothing(bra)
    if Person.rank_tits(the_person.tits) -1 > the_person.bra_size: #can't even fit in old tops
        $ top = test_outfit.get_upper_top_layer
        while top:
            $ test_outfit.remove_clothing(top)
            $ top = test_outfit.get_upper_top_layer
        $ test_outfit.add_upper(lab_coat.get_copy())
    $ the_person.apply_outfit(test_outfit)
    $ scene_manager.add_actor(the_person, test_outfit, position = "stand4", emotion = "sad")
    "As you are working one of your employees approaches, looking a little troubled."
    if Person.rank_tits(the_person.tits) -1 > the_person.bra_size:
        "You are a little surprised to see that she seems to be wearing a lab coat with no top on underneath."
    elif Person.rank_tits(the_person.tits) > the_person.bra_size:
        "It seems like she is not wearing a bra, and her breasts are straining to fit inside her top."
    if mc.location.person_count > 1:
        "She steps close, lowering her voice so that no one else can hear."
    the_person "Excuse me, [the_person.mc_title] I'm having a problem and was hoping to get some advice from you."
    mc.name "Sure, [the_person.title], what can I do for you?"
    if mc.location.person_count > 1:
        the_person "Could we go to your office so we can talk in private?"
        mc.name "Of course, lead the way."
        "You get up and follow her to your office, once there you take a seat and gesture for her to continue."
    the_person "This morning when I was getting dressed I was having some problems with my clothes."
    the_person "I tried a few different outfits, and it seems like nothing I have fits properly anymore."
    mc.name "What do you mean they don't fit?"
    if Person.rank_tits(the_person.tits) > the_person.bra_size:
        the_person "It's my breasts, they have grown a ridiculous amount. They don't fit in my bras anymore. I can't even get them on."
        mc.name "That sounds terrible."
        if Person.rank_tits(the_person.tits) -1 > the_person.bra_size:
            the_person "It is, in fact it's so bad I couldn't even get my top on. I had to grab a lab coat to wear since nothing else worked."
    elif Person.rank_tits(the_person.tits) < the_person.bra_size:
        the_person "It's my breasts, they seem to be shrinking. It's like going back before puberty."
        the_person "Suddenly all of my clothes are really loose."
        the_person "I didn't even wear a bra, because it just felt weird having the cups half full."
    "It sounds like your serums have had a profound effect on [the_person.title]. You better help her deal with the fall out of their effects."
    mc.name "This is serious, what can I do to help you?"
    the_person "I think I need to buy new clothes so that I can be comfortable and properly dressed for work."
    the_person "The problem is that I just paid rent and I can't afford a whole new wardrobe until my next paycheck."
    the_person "I was hoping I could get an advance or something so that I can go shopping tonight."
    "This is at least partially your fault, you should probably take responsibility for the problem."
    mc.name "I should be able to help you out."
    menu:
        "Give her an advance":
            mc.name "The company can certainly afford to pay you in advance. Your work is more than satisfactory."
            $ weeks_wages = the_person.current_job.salary*5
            $ scene_manager.update_actor(the_person, emotion = "happy")
            $ the_person.change_happiness(5 + mc.charisma)
            $ the_person.change_obedience(1 + mc.charisma)
            $ mc.business.funds -= weeks_wages
            $ add_unpaid_intern_clear_punishment_action(the_person.current_job, the_person.current_job.salary)
            $ the_person.current_job.salary -= the_person.current_job.salary
            "You pull out your wallet and start to pull out a few bills."
            "[the_person.title] takes the bills, then smiles broadly at you."
            the_person "That's very generous of you sir, thank you."
        "Give her a bonus":
            mc.name "In fact, you don't even need to consider it an advance."
            call employee_pay_cash_bonus(the_person) from _call_new_bra_bonus
    $ the_person.apply_outfit(test_outfit)
    $ scene_manager.update_actor(the_person)
    mc.name "Have you thought about what you want to buy when you go shopping?"
    the_person "I thought I would just buy the same clothes in a size that is more appropriate to my current situation."
    mc.name "That is one option, but why not take the opportunity to reassess your outfits. Are you happy with all of them the way they are?"
    the_person "I was so worried about getting something to wear I didn't consider that. Is there anything you think I should try and change?"
    $ alterations = 0
    if Person.rank_tits(the_person.tits) -1 > the_person.bra_size:
        mc.name "Well since it looks like you'll have to buy new shirts why don't we look at those first."
        the_person "Right, none of the ones I have fit right."
        menu:
            "Less conservative tops" if the_person.sluttiness > 30:
                mc.name "I think you could be a bit more daring. The clothes you have now are nice, but you might want to show off your new assets."
                the_person "You're right, it could be good to show off my body now that there is so much more to show."
                $ the_person.discover_opinion("skimpy outfits")
                $ alterations += smaller_shirts(the_person)
                if bra_count(the_person) == 1:
                    $ the_person.increase_opinion_score("skimpy outfits")
                if the_person.sluttiness > 50:
                    $ alterations += smaller_shirts(the_person)
                    if the_person.sluttiness > 70:
                        $ alterations += smaller_shirts(the_person)
                        if the_person.sluttiness > 90:
                            $ alterations += smaller_shirts(the_person)
            "The same tops":
                mc.name "I'm not sure if you'd be comfortable going with anything more daring that what you wear now."
                the_person "Yeah, I think that my current wardrobe is fine."
        mc.name "With that decided now we need to think about your bras."
    if Person.rank_tits(the_person.tits) > the_person.bra_size:
        mc.name "You will probably want some kind of support for your new breasts."
        if the_person.opinion.not_wearing_underwear > 0:
            the_person "Actually, I don't seem to have any stretch marks from their rapid expansion. I've never been a big fan of underwear anyway."
        elif the_person.opinion.not_wearing_underwear == 0:
            the_person "I suppose so, although I don't seem to have any stretch marks from their rapid expansion."
        else:
            the_person "Yes, I definitely want new bras, but surprisingly I don't seem to have any stretch marks to worry about."
    else:
        mc.name "Do you think you'll even want to wear bras, with tits that are so much smaller?"
        if the_person.opinion.not_wearing_underwear > 0:
            the_person "I don't know, I certainly didn't always wear bras. I kind of miss how free I was back then."
        elif the_person.opinion.not_wearing_underwear == 0:
            the_person "Maybe? I guess I just assumed I should wear bras, I've been doing it so long."
        else:
            the_person "Yes, I definitely want new bras, even if there isn't much to fill them."
    $ the_person.discover_opinion("not wearing underwear")
    "You think about her opinion, but since you are basically buying the clothes you might be able to push her outside of her comfort zone."
    menu:
        "No bras":
            mc.name "I want you to stop wearing bras."
            if the_person.opinion.not_wearing_underwear < 0:
                the_person "I'm not sure I'm comfortable with that."
                if the_person.sluttiness > 50 or the_person.obedience > 150 or the_person.love > 50:
                    mc.name "Don't worry [the_person.title] I'm comfortable with it, and isn't that what really matters?"
                    the_person "I guess you're right. I won't buy any new bras."
                    if bra_count(the_person) == 1:
                        $ the_person.increase_opinion_score("not wearing underwear")
                    $ alterations += bra_removal(the_person)
                else:
                    mc.name "I understand, you're not really ready for that yet."
            elif the_person.opinion.not_wearing_underwear == 0:
                the_person "Ok."
                if bra_count(the_person) == 1:
                    $ the_person.increase_opinion_score("not wearing underwear")
                $ alterations += bra_removal(the_person)
            else:
                if the_person.has_large_tits:
                    the_person "Oh, that sounds exciting, but my tits are so big."
                    mc.name "That's exactly why you should do it. Just imagine them swinging around all day for everyone to see."
                    the_person "That does sound pretty hot. I won't buy any new bras."
                elif the_person.tits in ["A", "AA"]:
                    the_person "That shouldn't be a problem. I hardly have anything to hold in them anyway."
                    mc.name "Exactly, and this way I'll be able to grab a little handful whenever I want."
                    the_person "I like the thought of that. I won't buy any new bras."
                else:
                    the_person "Ok."
                if bra_count(the_person) == 1:
                    $ the_person.increase_opinion_score("not wearing underwear")
                $ alterations += bra_removal(the_person)
        "Less conservative bras" if the_person.sluttiness > 30:
            mc.name "I want you to buy sexier bras."
            if the_person.opinion.not_wearing_underwear == 2:
                the_person "That is something I could probably do, but why don't I just stop wearing them altogether?"
                $ the_person.discover_opinion("not wearing underwear")
                menu:
                    "Yes":
                        mc.name "That would be even better. I guess you don't need to buy any new bras after all."
                        the_person "Yes, [the_person.mc_title]."
                        if bra_count(the_person) == 1:
                            $ the_person.increase_opinion_score("not wearing underwear")
                        $ alterations += bra_removal(the_person)
                    "No":
                        mc.name "No, I like you wearing them, I just want them to be skimpy and sexy."
                        the_person "I'll take a look and see what I can do for you."
            else:
                if the_person.opinion.lingerie > 0:
                    the_person "Well I do like having something sexy to show off when I take off my clothes."
                elif the_person.opinion.lingerie == 0:
                    the_person "Ok."
                else:
                    the_person "Underwear is underwear, does it really matter what it looks like?"
                    mc.name "It does for me, I love seeing your body highlighted by sexy clothes."
                    the_person "Alright, I'll try to find something more daring while I'm shopping."
            $ the_person.discover_opinion("lingerie")
            if bra_count(the_person) == 1:
                $ the_person.increase_opinion_score("skimpy outfits")
            $ alterations += sexier_bras(the_person)
            if the_person.sluttiness > 50:
                $ alterations += sexier_bras(the_person)
                if the_person.sluttiness > 70:
                    $ alterations += sexier_bras(the_person)
                    if the_person.sluttiness > 90:
                        $ alterations += sexier_bras(the_person)
        "The same bras":
            mc.name "What you already had was nice. Why don't you just get a better fitting for your new body?"
            the_person "Yes, that would probably be safest for now."
    if alterations > 10:
        the_person "You'll hardly recognise me with all the changes I'm making to my wardrobe."
        mc.name "I look forward to seeing the new you soon."
    elif alterations > 0 or bra_count(the_person) < 1:
        the_person "I guess I'll look a little different next time you see me."
        mc.name "I'm sure you'll look great in whatever you find."
    else:
        the_person "Thanks for the help, I can't wait to get back to my normal clothes."
        mc.name "Of course, let me know if you ever need more help in the future."
    if the_person.should_wear_uniform and Person.rank_tits(the_person.tits) -1 > the_person.bra_size:
        "With her problem solved, at least for now, you remember that her situation has forced her out of uniform."
        mc.name "It is understandable that you couldn't get into your uniform today, but it is still a serious violation of company policy."
        mc.name "I'm afraid I'm going to have to punish you for your failure to properly prepare for the day."
        the_person "Oh, right, I understand. Thank you for being so helpful. I'll definitely be in uniform the next time you see me."
        $ the_person.add_infraction(Infraction.out_of_uniform_factory())
    $ the_person.bra_size = Person.rank_tits(the_person.tits)
    $ scene_manager.clear_scene()
    return